Здравствуйте! Будьте добры, подскажите как исправить ситуацию:
Все работает, проблема в том, что последняя колонка Стоимость не заполняется, в силу того, что метод Загрузить() перезаписывает таблицу, где Стоимости нет. Как исправить? Не могу понять.

```1c
    &НаСервере
    Процедура ЗаполнитьДокументНаСервере() Экспорт

    Запрос = Новый Запрос;
    Запрос.Текст =
    "ВЫБРАТЬ
    | ТоварыНаСкладахОстатки.Номенклатура КАК Номенклатура,
    | ЕСТЬNULL(ТоварыНаСкладахОстатки.Характеристики, 0) КАК Характеристика,
    | ТоварыНаСкладахОстатки.КоличествоОстаток КАК Количество,
    | ЕСТЬNULL(ЦеныНоменклатурыСрезПоследних.Цена, 0) КАК ЦенаРеализации
    |ИЗ
    | РегистрНакопления.ТоварыНаСкладах.Остатки(
    | ,
    | Склад = &Склад
    | И Номенклатура.ВидНоменклатуры В (&ВидыТоваров)) КАК ТоварыНаСкладахОстатки
    | ЛЕВОЕ СОЕДИНЕНИЕ РегистрСведений.ЦеныНоменклатуры.СрезПоследних(, ВидЦены = &ВидЦены) КАК ЦеныНоменклатурыСрезПоследних
    | ПО ТоварыНаСкладахОстатки.Номенклатура = ЦеныНоменклатурыСрезПоследних.Номенклатура";

    ВидыТоваров = Новый Массив;
    ВидыТоваров.Добавить(Перечисления.ВидыНоменклатуры.ОсобыйТовар);
    ВидыТоваров.Добавить(Перечисления.ВидыНоменклатуры.Товар);

    Запрос.УстановитьПараметр("Склад",Склад);
    Запрос.УстановитьПараметр("ВидыТоваров",ВидыТоваров);
    Запрос.УстановитьПараметр("ВидЦены",ВидЦены);

    РезультатЗапроса = Запрос.Выполнить();
    ТЗ = РезультатЗапроса.Выгрузить();

    Для Каждого ТекСтрокаТовары Из Объект.Товары Цикл

    СтрокаТЗ = Объект.Товары.Добавить();
    СтрокаТЗ.Номенклатура = ТекСтрокаТовары.Номенклатура;
    СтрокаТЗ.Характеристика = ТекСтрокаТовары.Характеристика;
    СтрокаТЗ.Количество = ТекСтрокаТовары.Количество;
    СтрокаТЗ.ЦенаРеализации = ТекСтрокаТовары.ЦенаРеализации;
    СтрокаТЗ.Стоимость = СтрокаТЗ.Количество*СтрокаТЗ.ЦенаРеализации;

    КонецЦикла;
    Объект.Товары.Загрузить(ТЗ);
    //ЗаполнитьЗначенияСвойств(Объект.Товары[Объект.Товары.Количество()-1],ТЗ[0]);
    КонецПроцедуры
```
---
```1c
    Для Каждого ТекСтрокаТовары Из ТЗ Цикл

    СтрокаТЗ = Объект.Товары.Добавить();
    СтрокаТЗ.Номенклатура = ТекСтрокаТовары.Номенклатура;
    СтрокаТЗ.Характеристика = ТекСтрокаТовары.Характеристика;
    СтрокаТЗ.Количество = ТекСтрокаТовары.Количество;
    СтрокаТЗ.ЦенаРеализации = ТекСтрокаТовары.ЦенаРеализации;
    СтрокаТЗ.Стоимость = СтрокаТЗ.Количество*СтрокаТЗ.ЦенаРеализации;

    КонецЦикла;
    //Объект.Товары.Загрузить(ТЗ);
```

Или в запросе стоимость расчитывать
---
```1c
    РезультатЗапроса = Запрос.Выполнить();
    ТЗ = РезультатЗапроса.Выгрузить();
    Объект.Товары.Загрузить(ТЗ);
    Для Каждого ТекСтрокаТовары Из Объект.Товары Цикл

    ТекСтрокаТовары.Стоимость = ТекСтрокаТовары.Количество*ТекСтрокаТовары.ЦенаРеализации;

    КонецЦикла;
    // Объект.Товары.Загрузить(ТЗ);
```
---
```1c
    РезультатЗапроса = Запрос.Выполнить();
    ТЗ = РезультатЗапроса.Выгрузить();
    Объект.Товары.Загрузить(ТЗ);
    Для Каждого ТекСтрокаТовары Из Объект.Товары Цикл

    ТекСтрокаТовары.Стоимость = ТекСтрокаТовары.Количество*ТекСтрокаТовары.ЦенаРеализации;

    КонецЦикла;
    // Объект.Товары.Загрузить(ТЗ);
```
---
Спасибо!! Работают оба варианта! Я вам очень благодарна! Всем желаю крепкого здоровья в это непростое время и удачи во всем!
---
```1c
    Для Каждого ТекСтрокаТовары Из ТЗ Цикл

    СтрокаТЗ = Объект.Товары.Добавить();
    СтрокаТЗ.Номенклатура = ТекСтрокаТовары.Номенклатура;
    СтрокаТЗ.Характеристика = ТекСтрокаТовары.Характеристика;
    СтрокаТЗ.Количество = ТекСтрокаТовары.Количество;
    СтрокаТЗ.ЦенаРеализации = ТекСтрокаТовары.ЦенаРеализации;
    СтрокаТЗ.Стоимость = СтрокаТЗ.Количество*СтрокаТЗ.ЦенаРеализации;

    КонецЦикла;
    //Объект.Товары.Загрузить(ТЗ);
```

Или в запросе стоимость расчитывать
---
Ребята, спасибо большое! Это рабочий вариант. До этого делала выборку из ТЗ, но скорее всего метод лишний был...не заполнялось. А именно так работает! Так оперативно, большое спасибо за ответ!